

# Dynamic Structural Equation Modeling of Intensive Longitudinal Data Using Mplus Version 8 (Parts 2 and 3)

Bengt Muthén  
bmuthen@statmodel.com


Tihomir Asparouhov & Ellen Hamaker

PSMG talk, March 28, 2017

Part 4 is presented by Tihomir Asparouhov, April 25

We thank Noah Hastings for excellent assistance

# Previously on DSEM at PSMG: A Paradigm Shift (Hamaker 3/14/17)



Source: Hamaker & Wichers (2017). No time like the present...  
*Current Directions in Psychological Science*.

## Different forms of intensive longitudinal data (ILD):

- daily diary (DD); self-report end-of-day
- experience sampling method (ESM); self-report of subjective experience
- ecological momentary assessment (EMA); healthcare related self-report
- ambulatory assessment (AA); physiological measurements
- event-based measurements; self-report after a particular event
- observational measurements; expert rater

## For more info on **methodology**, check out:

- Seminar of Tamlin Conner and Joshua Smyth on YouTube (<https://www.youtube.com/watch?v=nQBBVp9vBIQ>)
- Society for Ambulatory Assessment (<http://www.saa2009.org/>)
- Life Data (<https://www.lifedatacorp.com/>)
- Quantified Self (<http://quantifiedself.com/>)

## Data structure:

- one or more measurements per day
- typically for multiple days
- sometimes multiple waves (i.e., Nesselroade's measurement-burst design)

## Advantages of ESM, EMA and AA

- no recall bias
- high ecological validity
- physiological measures over a large time span
- monitoring of symptoms and behavior, with new possibilities for feedback and intervention (e-Health and m-Health)
- window into the dynamics of processes

## Methods for Analyzing Intensive Longitudinal Data

- Time series analysis ( $N = 1$ )
- Two-level time series analysis ( $N > 1$ )
  - Random effects varying across subjects (subject is level 2, so many more random effects than usual)
- Cross-classified time series analysis
  - Random effects varying across subjects and time
- Dynamic Structural Equation Modeling (DSEM)
  - General latent variable modeling
  - Bayesian estimation
  - Statistical background:
    - Asparouhov, Hamaker & Muthén (in preparation). Dynamic structural equation models
    - Asparouhov, Hamaker & Muthén (2017). Dynamic latent class analysis. *Structural Equation Modeling*, 24, 257-269

The Version 8 Mplus User's Guide adds  $N=1$  examples 6.23 - 6.28 and  $N > 1$  examples 9.30 - 9.40, many with two parts (basic and advanced).

# Overview

- Example: Smoking cessation (EMA)
- Prelude 1: Brief overview of some familiar methods for longitudinal data
  - Why is regular growth modeling not sufficient for ILD?
- Prelude 2: Bayesian analysis - a thumbnail sketch
- Smoking data applications:
  - $N = 1$  time series analysis
  - Two-level time series analysis
  - Cross-classified time series analysis - looking for trends over time
  - Adding trend to two-level time series analysis
  - Cross-classified time series analysis with a trend
- Time-varying effect modeling (TVEM) using cross-classified time series analysis
- Miscellaneous:
  - Latent variable models
  - Upcoming talks and workshops on DSEM
  - Non-time series news in Mplus version 8

- Shiffman smoking cessation data
- $N = 230$ ,  $T \approx 150$ : Random prompts from Personal Digital Assistant (hand held PC) approx. 5 times per day for a month
- Variables: Smoking urge (0-10 scale), negative affect (unhappy, irritable, miserable, tense, discontent, frustrated-angry, sad), gender, age, quit/relapse
- Shiyko et al. (2012). Using the time-varying effect model (TVEM) to examine dynamic associations between negative affect and self confidence on smoking urges. *Prevention Science*, 13, 288-299


- Non-intensive longitudinal data:
  - $T$  small (2 - 10) and  $N$  large
  - Modeling: Auto-regressive (cross-lagged) and growth modeling
- Intensive longitudinal data:
  - $T$  large (30-200) and  $N$  smallish (even  $N = 1$ ) but can be 1,000.  
Often  $T > N$
  - Modeling: We shall see




# Common Methods for Non-Intensive Longitudinal Data

$N$  large and  $T$  small (2 - 10):

## (1) Auto-Regressive Modeling





Cross-lagged modeling (e.g.  $y$  = urge,  $z$  = negative affect):



Extensions of the classic cross-lagged panel model:

- Hamaker et al., *Psych Methods* 2015: The random intercepts cross-lagged panel model
- Curran et al., *J of Consulting & Clinical Psych* 2014: The separation of between-person and within-person components
- Berry and Willoughby, *Child Development* 2016: Rethinking the cross-lagged panel model (growth model added)
  - Both models are fitted in Mplus


# Common Methods for Non-Intensive Longitudinal Data: (2) Growth Modeling



There are 2 problems:


- 1 Correlation between time points not fully explained by growth factors alone due to closely spaced measurements - autocorrelation needs to be added
- 2 Time series are too long due to slow computations

# Solving Problem 1. Add Residual (Auto) Correlation: Growth Modeling In Single-Level, Wide Format Version y as 5 columns in the data




Mplus User's Guide ex6.17 - but cumbersome with large T.

# Solving Problem 2. Switch From Single-Level to Two-Level, Long Format Version: y as 1 column in the data



Within (level-1)  
Variation across time

-----  
Between (level-2)  
Variation across subject




Mplus User's Guide ex9.16

# Growth Modeling In Two-Level, Long Format



Within (level-1)  
Variation across time

Between (level-2)  
Variation across subject





---

```
VARIABLE: CLUSTER = subject;  
          WITHIN = time;  
          BETWEEN = w;  
ANALYSIS: TYPE = TWOLEVEL RANDOM;  
MODEL:   %WITHIN%  
         s | y ON time;  
         %BETWEEN%  
         y s ON w; ! y is the same as i  
         y WITH s;
```

---

But where is the autocorrelation? And how can it be made random?

# Solution: Two-Level Time Series Analysis With A Trend Allowing Autocorrelation and Many Time points



Autoregression for the residuals instead?

Hamaker (2005). Conditions for the equivalence of the autoregressive latent trajectory model and a latent growth curve model with autoregressive disturbances. *Sociological Methods & Research*.



- Bayesian advantages over ML
- Convergence of Bayes iterations
- Trace and autocorrelation plots
- Speed of Bayes in Mplus
- Bayes references

# Bayesian Analysis: Advantages over ML

- Six key advantages of Bayesian analysis over frequentist analysis using maximum likelihood estimation:
  - ➊ More can be learned about parameter estimates and model fit
  - ➋ Small-sample performance is better and large-sample theory is not needed
  - ➌ Parameter priors can better reflect results of previous studies
  - ➍ Analyses are in some cases less computationally demanding, for example, when maximum-likelihood requires high-dimensional numerical integration
  - ➎ In cases where maximum-likelihood computations are too computationally demanding, Bayes with non-informative priors can be viewed as a computing algorithm that would give essentially the same results as maximum-likelihood if maximum-likelihood estimation were computationally feasible
  - ➏ New types of models can be analyzed where the maximum-likelihood approach is impossible


# Why Are Bayesian Computations Possible Where ML Computations Are Not?

The general modeling features of DSEM make ML almost impossible, creating the need for Bayesian estimation.

An intuitive description of the computational difference between ML and Bayes (with non-informative priors):

- ML works with the joint distribution of all variables whereas Bayes works with a series of conditional distributions
- The joint distribution can be difficult to describe whereas the conditional distributions can be easier
- Bayes is sometimes the only feasible alternative when the joint distribution is hard to formulate

Figure : Prior, likelihood, and posterior for a parameter





- Priors:

- Non-informative priors (diffuse priors): Large variance (default in Mplus)
  - A large variance reflects large uncertainty in the parameter value. As the prior variance increases, the Bayesian estimate gets closer to the maximum-likelihood estimate
- Weakly informative priors: Used for technical assistance
- Informative priors:
  - Informative priors reflect prior beliefs in likely parameter values
  - These beliefs may come from substantive theory combined with previous studies of similar populations


# Bayes' Advantage Over ML: Informative Priors

- Frequentists often object to Bayes using informative priors
- But they already do use such priors in many cases in unrealistic ways (e.g. factor loadings fixed exactly at zero)
- Bayes can let informative priors reflect prior studies
- Bayes can let informative priors identify models that are unidentified by ML which is useful for model modification (BSEM)
- The credibility interval for the posterior distribution is narrower with informative priors

# Bayes Posterior Distribution Similar to ML Bootstrap Distribution: Credibility versus Confidence Intervals



# Convergence: Trace Plot for Two MCMC Chains. PSR




Potential scale reduction criterion (Gelman & Rubin, 1992):

$$PSR = \sqrt{\frac{W+B}{W}}, \quad (1)$$

where  $W$  represents the within-chain variation of a parameter and  $B$  represents the between-chain variation of a parameter. A PSR value close to 1 means that the between-chain variation is small relative to the within-chain variation and is considered evidence of convergence.

# Convergence of the Bayes Markov Chain Monte Carlo (MCMC) Algorithm

**Figure :** Premature stoppage of Bayes MCMC iterations using the Potential Scale Reduction (PSR) criterion







# TECH8 Screen Printing of Bayes MCMC Iterations

between = female age;  
with;  
lags;  
miss;  
tinte;  
useob;  
ar  
ar  
ar  
ar  
ar  
ar  
idvar;  
ar  
Define:  
femal  
age =  
cente  
Analysis:  
type  
estin  
proc  
bite:  
Model:  
%wit:  
resid;  
phi | resid on resid1;  
urge@0.01;  
logv | resid;  
syx | resid on negaff;  
negaff;  
  
%Between%  
urge phi logv syx on female age;  
urge phi logv syx with urge phi logv syx;

ITERATION	POTENTIAL SCALE REDUCTION	PARAMETER WITH HIGHEST PSR	TIME	TOTAL TIME
100	1.080	12	16.83	16.8
200	1.072	20	17.27	34.1
300	1.057	19	16.78	50.9
400	1.085	3	16.71	67.6
500	1.076	16	17.07	84.7
600	1.082	21	16.77	101.4
700	1.098	23	16.99	118.4
800	1.112	20	16.80	135.2
900	1.072	23	16.79	152.0
1000	1.063	8	16.97	169.0
1100	1.106	8	16.90	185.9
1200	1.149	8	16.97	202.9
1300	1.179	8	17.04	219.9
1400	1.182	8	17.23	237.1
1500	1.210	8	16.87	254.0
1600	1.214	8	18.05	272.1
1700	1.206	8	17.40	289.5
1800	1.178	8	17.40	306.9
1900	1.158	8	17.28	324.1
2000	1.144	8	17.09	341.2
2100	1.131	8	16.80	358.0

to finish its execution...  
-C in the MS-DOS wind

# Trace and Autocorrelation Plots Indicating Poor Mixing



Wang & Preacher (2014). Moderated mediation analysis using Bayesian methods. *Structural Equation Modeling*.

- Comparison of ML (with bootstrap) and Bayes: Similar statistical performance
- Comparison of Bayes using BUGS versus Mplus: Mplus is 15 times faster
- Reason for Bayes being faster in Mplus:
  - Mplus uses Fortran (fastest computational environment)
  - Mplus uses parallel computing so each chain is computed separately
  - Mplus uses the largest updating blocks possible - complicated to program but gives the best mixing quality
  - Mplus uses sufficient statistics when possible
- Mplus Bayes considerably easier to use

# Nevertheless - It's Going To Be Slower Than Usual: Timings For The Runs In This Talk

Using smoking data with  $N = 230$ ,  $T \approx 150$

- N=1 analysis of subject 227: 0 seconds
- First two-level analysis: 2:37
- Two-level regression analysis: 3:06
- Cross-classified analysis: 19:36
- Two-level trend analysis: 4:01
- Cross-classified trend analysis: 33:11

Bengts PC as of June 2012: Dell XPS 8500, i7-3770 with 8 processors, CPU of 3.40 GHz, 12 GB RAM, 64-bit.


- Gelman et al. (2014). *Bayesian Data Analysis*, 3rd edition
- Lynch (2010). *Introduction to Applied Bayesian Statistics and Estimation for Social Scientists*
- Bayes technical reports on the Mplus website: See [www.statmodel.com](http://www.statmodel.com) under Papers, Bayesian Analysis
- Muthén (2010). Bayesian analysis in Mplus: A brief introduction. Technical Report. [www.statmodel.com](http://www.statmodel.com)
- Chapter 9 of Muthén, Muthén & Asparouhov (2016). *Regression and Mediation Analysis using Mplus*

- Shiffman smoking cessation data
- N = 230, T  $\approx$  150: Random prompts from Personal Digital Assistant (hand held PC) approx. 5 times per day for a month
- Variables: Smoking urge (1-11 scale), negative affect (unhappy, irritable, miserable, tense, discontent, frustrated-angry, sad), gender, age, quit/relapse


Replicated time series analysis of N = 1 is possible using the R package MplusAutomation by Hallquist (Schultzberg; Uppsala University Statistics Department).

# N = 1 Time Series Analysis Of Subjects 227 And 5

Smoking urge plotted against time for subject 227 (didn't quit)



Smoking urge plotted against time for subject 5 (did quit)



# N = 1 Time Series Analysis Of Subjects 227 And 5

	Estimate	Posterior S.D.	One-Tailed P-Value	95% C.I.		Significance
				Lower 2.5%	Upper 2.5%	
urge ON						
urge&1	0.112	0.068	0.060	-0.027	0.240	
negaff	1.196	0.178	0.000	0.810	1.542	*
Intercepts						
urge	4.882	0.494	0.000	3.899	5.865	*
Residual Variances						
urge	5.719	0.635	0.000	4.646	7.070	*

	Estimate	Posterior S.D.	One-Tailed P-Value	95% C.I.		Significance
				Lower 2.5%	Upper 2.5%	
urge ON						
urge&1	0.822	0.050	0.000	0.723	0.918	*
negaff	-0.257	0.408	0.272	-1.087	0.516	
Intercepts						
urge	0.517	0.377	0.074	-0.247	1.230	
Residual Variances						
urge	2.007	0.272	0.000	1.566	2.617	*




- Used to create a new time variable and insert missing data records when data are misaligned with respect to time:
  - due to missed measurement occasions that are not assigned a missing value flag
  - due to random measurement occasions

For more details, technical discussion and simulations, see Asparouhov, Hamaker, Muthén (2017) in preparation.

# N = 1 Time Series Analysis Using $T_{interval} = \text{timeqd}(0.08)$

Subject 5 (did quit):  $T_{interval}$  results in missing data records inserted to resolve different time distances between measurements



	Estimate	Posterior S.D.	One-Tailed P-Value	95% C.I.		Significance
				Lower 2.5%	Upper 2.5%	
<b>Subject 5 without <math>T_{interval}</math></b>						
urge ON						
urge&1	0.822	0.050	0.000	0.723	0.918	*
negaff	-0.257	0.408	0.272	-1.087	0.516	
<b>Subject 5 with <math>T_{interval}</math></b>						
urge ON						
urge&1	0.844	0.037	0.000	0.772	0.917	*
negaff	-0.158	0.382	0.328	-0.930	0.577	

# Mplus Input For Subject 5 Time Series Regression


---

TITLE: Shiffman smoking urge data N = 1 model for subject 5 (quit=1)  
DATA: FILE = combined\_relapsers\_quitters\_03-17-17.csv;  
VARIABLE: NAMES = subject t day urge craving negaff arousal timeqd  
gender age quit;  
! quit = 1 for quitters, 0 for relapsers  
USEVARIABLES = urge negaff;  
**LAGGED = urge(1);**  
MISSING = ALL(999);  
**TINTERVAL = timeqd(0.08);**  
**USEOBSERVATIONS = subject EQ 5;**  
IDVARIABLE = \_recnum;  
ANALYSIS: **ESTIMATOR = BAYES;**  
PROCESSORS = 2;  
BITERATIONS = (1000);  
MODEL: **urge ON urge&1 negaff;**  
negaff;  
OUTPUT: **TECH1 TECH8 STANDARDIZED TECH4 RESIDUAL;**  
PLOT: TYPE = PLOT3;

---

Analysis of all  $N = 230$  smoking data subjects, allowing for parameter variation across subjects

# Univariate Analysis of Smoking Urge Data: Two-Level Time Series Model



VARIABLE: NAMES = subject t day urge craving negaff arousal timeqd  
gender age quit;  
!quit = 1 for quitters, 0 for relapsers  
USEVARIABLES = urge quit age female;  
**CLUSTER = subject;**  
**BETWEEN = female age quit;**  
CATEGORICAL = quit;  
LAGGED = urge(1);  
MISSING = ALL(999);  
TINTERVAL = timeqd(0.08);  
IDVARIABLE = \_recnum;

DEFINE: female = gender - 1;  
age = (age-44.3)/10.1;

ANALYSIS: **TYPE = TWOLEVEL RANDOM;**  
ESTIMATOR = BAYES;  
PROCESSORS = 2;  
BITERATIONS = (1000);

---

# Input Continued

---


MODEL:            %WITHIN%  
                  **phi | urge ON urge&1;**  
                  **logv | urge;**  
                  %BETWEEN%  
                  urge phi logv ON female age;  
                  urge phi logv WITH urge phi logv;  
                  quit ON urge phi logv female age;

OUTPUT:           TECH1 TECH8 FSCOMPARISON STANDARDIZED  
                  TECH4 RESIDUAL;

PLOT:             TYPE = PLOT3;  
                  **FACTORS = ALL;**

---


# Results for Univariate Analysis of Smoking Urge Data: Two-Level Time Series Model - Between Part



- Random effects regressed on female:
  - females have lower autocorrelation (the other way around for NA)
  - females have higher residual variance
- Quit (binary) regressed on random effects:
  - higher urge gives lower quit probability
  - higher autocorrelation gives higher quit probability
  - higher residual variance gives lower quit probability



# Results for Univariate Analysis of Smoking Urge Data: Two-Level Time Series Model - Between Part




- Indirect effects need counterfactual definition due to the binary quit variable (counterfactually-defined causal effects on the between level)
- For references, see the Mplus Mediation web page <http://www.statmodel.com/Mediation.shtml>

# How Big Does N and T Need to Be?

## Quality of Estimation and Power to Detect Effects

- Simulation study (Schultzberg; Uppsala University Statistics Department): presentation at Mplus User's Meeting and at IMPS
- Different needs for different relations - random effects as predictors hardest
- For certain models satisfactory results can be obtained also for short time series such as  $T = 10$  (or even 5)
  - Bayes versus ML for short time series? Paul Allison on dynamic panel data modeling using ML; better than econometric AB estimation





# Technical Interlude On How To Draw Model Diagrams




Within




Between



Between



# Regression Analysis of Smoking Urge Data: Two-Level Time Series Model Adding Negative Affect (na)



Within

---

```
VARIABLE:      NAMES = subject t day urge craving negaff arousal timeqd
                gender age quit;
                !quit = 1 for quitters, 0 for relapsers
                USEVARIABLES = urge negaff age female;
                CLUSTER = subject;
                BETWEEN = female age;
                WITHIN = negaff;
                LAGGED = urge(1);
                MISSING = ALL(999);
                TINTERVAL = timeqd(0.08);

DEFINE:        female = gender - 1;
                age = (age-44.3)/10.1;
                CENTER negaff(GROUPMEAN);

ANALYSIS:      TYPE = TWOLEVEL RANDOM;
                ESTIMATOR = BAYES;
                PROCESSORS = 2;
                BITERATIONS = (1000);
```

---

---

MODEL:            %WITHIN%  
                  phi | urge ON urge&1;  
                  logv | urge;  
                  **syx | urge ON negaff;**  
                  negaff;  
                  %BETWEEN%  
                  urge phi logv syx ON female age;  
                  urge phi logv syx WITH urge phi logv syx;


OUTPUT:           TECH1 TECH8 FSCOMPARISON STANDARDIZED  
                  TECH4 RESIDUAL;

PLOT:             TYPE = PLOT3;  
                  FACTORS = ALL;


---

# Technical Interlude

## On Where The Autocorrelation Should Be Applied



Within



Within

# Mplus Input for AR(1) for the Residuals

---

```
VARIABLE:      NAMES = subject t day urge craving negaff arousal timeqd
                gender age quit;
                !quit = 1 for quitters, 0 for relapsers
                USEVARIABLES = urge negaff age female;
                CLUSTER = subject;
                BETWEEN = female age;
                WITHIN = negaff;
                ! LAGGED = urge(1);
                MISSING = ALL(999);
                TINTERVAL = timeqd(0.08);

DEFINE:        female = gender - 1;
                age = (age-44.3)/10.1;
                CENTER negaff(GROUPMEAN);

ANALYSIS:      TYPE = TWOLEVEL RANDOM;
                ESTIMATOR = BAYES;
                PROCESSORS = 2;
                BITERATIONS = (10000);
```

---



---

MODEL:            %WITHIN%  
                  ! define the residual as a factor:  
                  **resid BY urge (&1);**  
                  **phi | resid ON resid&1;**  
                  **urge@0.1;**  
                  logv | resid;  
                  syx | urge ON negaff;  
                  negaff;  
                  %BETWEEN%  
                  urge phi logv syx ON female age;  
                  urge phi logv syx WITH urge phi logv syx;

OUTPUT:            TECH1 TECH8 FSCOMPARISON STANDARDIZED TECH4  
                  RESIDUAL;

PLOT:              TYPE = PLOT3;  
                  FACTORS = ALL;

---

Cross-classified analysis uses two between-level cluster variables: subject crossed with time (one observation for a given subject at a given time point).  $N > 1$

The cross-classified model is a generalization of the two-level model and provides more flexibility in that random effects can vary across not only subject but also time. The Bayes MCMC algorithm is more complex and considerably slower.


Consider the two-level model with a random intercept/mean:

$$y_{it} = \alpha + \alpha_i + \beta y_{w,it-1} + \text{residual}. \quad (2)$$


The corresponding cross-classified model is:

$$y_{it} = \alpha + \alpha_i + \alpha_t + \beta y_{w,it-1} + \text{residual}. \quad (3)$$

# Regression Analysis of Smoking Urge Data: Cross-Classified Time Series Model Adding Variation Across Time



Within



Between Subject

Between Time



VARIABLE: NAMES = subject t day urge craving negaff arousal timeqd  
gender age quit;  
!quit = 1 for quitters, 0 for relapsers  
USEVARIABLES = urge quit negaff age female;  
**CLUSTER = subject timeqd;**  
**BETWEEN = (subject) female age quit;**  
CATEGORICAL = quit;  
WITHIN = negaff;  
LAGGED = urge(1);  
MISSING = ALL(999);  
TINTERVAL = timeqd(0.08);

DEFINE: female = gender - 1;  
age = (age-44.3)/10.1;  
CENTER negaff(GROUPMEAN SUBJECT);

ANALYSIS: **TYPE = CROSS RANDOM;**  
ESTIMATOR = BAYES;  
PROCESSORS = 2;  
BITERATIONS = (1000);

---

---


```
MODEL:          %WITHIN%  
               urge ON urge&1;  
               syx | urge ON negaff;  
               negaff;  
               %BETWEEN subject%  
               urge syx ON female age;  
               urge WITH syx;  
               quit ON urge syx female age;  
               %BETWEEN timeqd%  
               urge WITH syx;  
OUTPUT:        TECH1 TECH8;  
PLOT:          TYPE = PLOT3;  
              FACTORS = ALL;
```

---

Run time: 19:36. Using fixed syx slope takes only 2 minutes.


# Regression Analysis of Smoking Urge Data: Cross-Classified Time Series Model

## Urge Factor Score Plotted Against Time for All Subjects



# Regression Analysis of Smoking Urge Data: Cross-Classified Time Series Model

## Urge on Na Slope (syx) Factor Score Plotted Against Time




# Cross-Classified Analysis: a Quick Way to Spot a Trend in a Variable

Does negative affect show a trend? Run time is only 1:22 for fixed AR(1):


```
MODEL:          %WITHIN%  
                negaff ON negaff&1;  
                %BETWEEN subject%  
                negaff;  
                %BETWEEN timeqd%  
                negaff;
```

Mean negaff factor score plotted against time:





# Modeling the Trend: Recall How Growth Modeling Can Be Transformed From Wide To Long




# Growth Modeling: Two-Level, Long Format Version



Within (level-1)  
Variation across time


Between (level-2)  
Variation across individual



# Two-Level Time Series Analysis of Smoking Urge data

## Adding a Trend for Urge.

### - Growth Analysis with a Time-Varying Covariate



- Interpretation of  $s$  not the usual one; direct effect at each time
- An alternative formulation places the autoregression on the residuals (Hamaker, 2005; *SM&R*), resulting in the usual  $s$  interpretation

# Mplus Input For Two-Level Trend Analysis

---

VARIABLE: NAMES = subject t day urge craving negaff arousal timeqd  
gender age quit;  
USEVARIABLES = urge quit negaff age female time;  
CLUSTER = subject;  
BETWEEN = female age quit;  
CATEGORICAL = quit;  
**WITHIN = time negaff;**  
LAGGED = urge(1);  
MISSING = ALL(999);  
TINTERVAL = timeqd(0.08);

DEFINE: female = gender - 1;  
age = (age-44.3)/10.1;  
**time = (timeqd/10-10.49)/6.14;**  
CENTER negaff(GROUPMEAN);

ANALYSIS: TYPE = TWOLEVEL RANDOM;  
ESTIMATOR = BAYES;  
PROCESSORS = 2;  
BITERATIONS = (1000);

---

---

MODEL:            %WITHIN%  
                 phi | urge ON urge&1;  
                 syx | urge ON negaff;  
                 logv | urge;  
                 **s | urge ON time;**  
                 negaff; time;  
                 %BETWEEN%  
                 urge syx s phi logv ON female age;  
                 urge syx s phi logv WITH urge syx s phi logv;  
                 quit ON urge syx s phi logv female age;


OUTPUT:           TECH1 TECH8 FSCOMPARISON STANDARDIZED TECH4  
                 RESIDUAL;

PLOT:             TYPE = PLOT3;  
                 FACTORS = ALL;


---

Run time: 4:19

# Results for Two-Level Regression Analysis of Smoking Urge Data: Adding a Trend for Urge. - Growth Analysis with a Time-Varying Covariate



# Cross-Classified Regression Analysis of Smoking Urge Data: Adding a Trend for Urge and the Regression Slope Cross-Classified Growth Analysis With a Time-Varying Covariate



# Mplus Input for Cross-Classified Regression Analysis with a Trend

---

VARIABLE: NAMES = subject t day urge craving negaff arousal timeqd  
gender age quit;  
! quit = 1 for quitters, 0 for relapsers  
USEVARIABLES = urge quit negaff age female time timet;  
CLUSTER = subject timeqd;  
**BETWEEN = (subject) female age quit (timeqd) timet;**  
CATEGORICAL = quit;  
**WITHIN = negaff time;**  
LAGGED = urge(1);  
MISSING = ALL(999);  
TINTERVAL = timeqd(0.08);

DEFINE: female = gender - 1;  
age = (age-44.3)/10.1;  
time = (timeqd/10-10.49)/6.14;  
**timet = time;**

ANALYSIS: TYPE = CROSS RANDOM;  
ESTIMATOR = BAYES;  
PROCESSORS = 2;  
BITERATIONS = (1000);

---



# Input Continued

---

MODEL:            % **WITHIN** %  
                  urge ON urge&1;  
                  syx | urge ON negaff;  
                  **s | urge ON time;**  
                  negaff;  
                  time;  
                  % **BETWEEN subject** %  
                  urge syx s ON female age;  
                  urge syx s **WITH** urge syx s;  
                  quit ON urge syx s female age;  
                  % **BETWEEN timeqd** %  
                  **syx ON timet;**  
                  urge **WITH** syx;  
                  s@0;  
OUTPUT:           TECH1 TECH8;  
PLOT:             TYPE = PLOT3;  
                  FACTORS = ALL;

---

# Time-Varying Effect Modeling (TVEM) Compared To Cross-Classified DSEM

For individual  $i$  at the  $j^{\text{th}}$  observation,

$$y_{ij} = \beta_0(t_{ij}) + \beta_1(t_{ij}) x_{ij} + \varepsilon_{ij},$$


where  $\beta_0(t_{ij})$  and  $\beta_1(t_{ij})$  are continuous functions of time using P-spline-based methods and varying the number of knots (Hastie & Tibshirani, 1990).

Based on the varying coefficient model of Hastie & Tibshirani (1993) in *Journal of the Royal Statistical Society, Series B*.


# TVEM Example: Smoking Cessation Study

- Shiyko et al. (2012). Using the time-varying effect model (TVEM) to examine dynamic associations between negative affect and self confidence on smoking urges. *Prevention Science*, 13, 288-299



Recall our two-level time series findings for these data:



# Shiyko et al. (2012) TVEM Results for the Smoking Urge Intercept




# Shiyko et al. (2012) TVEM Results for the Slopes: Negative Affect and Confidence as Smoking Urge Predictors



# Advantages of DSEM over TVEM

- TVEM: Time-varying effect model
  - Regression analysis with time-varying coefficients using splines
- DSEM:
  - Richer set of models: Multivariate, latent variables, auto-regression, variance modeling
  - Predictors and distal outcomes of subject- and time-varying parameters (e.g. female  $\rightarrow$  phi/logv  $\rightarrow$  quit)
  - Functional form informed by the data: Estimated by random effect scores as shown below for smoking urge
  - Growth trend easily added




# Time Series Analysis with Latent Variables: Latent Variables on the Within Level

- So far we have focused on latent variables on the between level in the form of random effects
  - Although on Within we have used the latent variable within-level decomposition of the outcome, centering by  $y_{bi}$ :

$$y_{wit} = y_{it} - y_{bi}$$

- Now we introduce within-level factors:
  - Factors defined by single indicators with measurement error
  - Residual factors in ARMA(1,1)
  - Factors defined by multiple indicators
  - Two-level and Cross-classified analysis
- Categorical latent variables (version 8.1, although an SEM article is already online; Asparouhov, Hamaker, Muthén, 2017):
  - Transition modeling (Hidden Markov, regime switching, time-series LTA) with latent class variables
  - Growth mixture modeling

# Single-Indicator Measurement Error Model versus ARMA(1,1): Measurement Error Model



Within

Between

---

TITLE: this is an example of a two-level time series analysis with a first-order autoregressive AR(1) factor analysis model for a single continuous indicator and measurement error

DATA: FILE = ex9.33.dat;

VARIABLE: NAMES = y subject;  
CLUSTER = subject;

ANALYSIS: TYPE = TWOLEVEL RANDOM;  
ESTIMATOR = BAYES;  
PROCESSORS = 2;  
BITERATIONS = (5000);

MODEL: %WITHIN%  
**f BY y@1(&1);**  
**s | f ON f&1;**  
%BETWEEN%  
y WITH s;


OUTPUT: TECH1 TECH8;

PLOT: TYPE = PLOT3;

---



# ARMA(1,1)




---

MODEL:    %WITHIN%  
          s | y ON y&1;  
          e BY y@1 (&1);  
          y@.01;  
          y ON e&1;

---

Within


Between



- AR stands for autoregressive and MA stands for moving average (Shumway & Stoffer, 2011)


- Granger and Morris (1976) and Schuurman et al. (2015) show that for  $N = 1$  time series analysis, ARMA (1, 1) is an alternative representation of the data used in the measurement error model; formulas show translation of parameters
  - In the Mplus implementation the measurement error formulation converges more smoothly than ARMA(1,1)
  - The  $N = 1$  versions of these models require a large  $T$ , say  $T > 100$
  - Preliminary simulations indicate that the  $N > 1$  versions have good performance at  $T = 50$ , reasonable performance at  $T = 25$ , and maybe acceptable performance at  $T = 14$ : Suitable for daily diary designs
- AR models assume exponential decays in autocorrelation - the measurement error model allows a slower, more realistic decay (Asparouhov, 2017)
- A preliminary observation: it appears to be difficult to add random variance to the factor in the measurement error model
- Research questions: How does performance compare to having multiple indicators (e.g. 10 NA items)? Is random variance easier there?

# Two-Level Factor Analysis: UG ex9.34




- DAFS (direct autoregressive factor score) versus WNFS (white noise factor score) modeling
- N = 1 factor analysis: Engle & Watson (1981) in JASA, Molenaar (1985) in Psychometrika

# Cross-Classified Factor Analysis: UG ex9.40




Within

Between  
subject



Between  
time



# Two Consequences of Two-Level and Cross-Classified Factor Analysis: (1) Allowing Measurement Non-Invariance Across Subjects and Time

For a certain item measured for individual  $i$  at time  $t$ , two-level factor analysis (see, e.g., Muthén, 1994) considers

$$y_{it} = \nu + \lambda_b f_{bi} + \varepsilon_{bi} + \lambda_w f_{wit} + \varepsilon_{wit}. \quad (4)$$

This can be re-expressed as

$$\text{Level 1 : } y_{it} = \nu_i + \lambda_w f_{wit} + \varepsilon_{wit}, \quad (5)$$

$$\text{Level 2 : } \nu_i = \nu + \lambda_b f_{bi} + \varepsilon_{bi}, \quad (6)$$

which is a random intercept model, that is, there is measurement non-invariance across subjects wrt the intercepts (Muthén & Asparouhov, 2017). Measurement non-invariance of intercepts across time and due to random loadings can also be studied using cross-classified analysis (Asparouhov & Muthén, 2015; Fox, 2010). For random loadings varying across subjects, see the V8 User's Guide ex9.40 part2.

IRT often considers the special case of measurement invariance with equal loadings across levels,  $\lambda_b = \lambda_w$ , and zero between-level residuals which can be expressed as

$$\text{Level 1 : } y_{it} = \nu + \lambda f_{it} + \varepsilon_{wit}, \quad (7)$$

$$\text{Level 2 : } f_{it} = f_{bi} + f_{wit}. \quad (8)$$

# Two Consequences of Two-Level and Cross-Classified Factor Analysis: (2) Subject-Specific Reliability

- The two-level and cross-classified factor analysis models imply
  - Measurement intercept and loadings possibly varying across subject and time
  - Factor variances and residual variances varying across subject and time
- This implies that reliabilities of test scores (based on a set of items) vary across subject and time
  - Hu, Nesselroade et al. (2016). Test reliability at the individual level. *Structural Equation Modeling*.
- But why not instead look at the precision with which the factor scores can be estimated?
  - Mplus Version 8 Monte Carlo simulations give correlations between true scores and estimated scores


# A Second Example

## A Closer Look at Negative Affect: Item Factor Analysis


- Data from the older cohort of the Notre Dame Study of Health & Well-being (Bergeman):  $N = 270$ ,  $T=56$  (daily measures on consecutive days)
- Wang, Hamaker, Bergeman (2012). Investigating inter-individual differences in short-term intra-individual variability. *Psychological Methods*
- Predictors and distal outcomes of negative affect development over the 56 days
- 10 NA items (5-cat scale): afraid, ashamed, guilty, hostile, scared, upset, irritable, jittery, nervous, distressed (average score used in article)
- Question format: Today I felt... (1 = Not at all, ..., 5 = Extremely)

# Negative Affect Distributions of NA in Bergeman Data

Average score (55% at floor value of 1 - Not at all for all 10 items):




Typical item distribution (66% at lowest value - Not at all):





# Ordered Categorical Item Modeling: Proportional Odds Model (Graded Response Model)



- Despite non-normal  $y$ , we can have normality of:
  - The latent response variable  $y^*$
  - Any factors in the model
  - The between-level random effects

# Mplus Input for Cross-Classified Factor Analysis with One Factor for 10 Ordinal NA Items

---

TITLE: Bergeman twolevel  
DATA: FILE = bergeman.csv;  
VARIABLE: NAMES = subject gender age hosp1 chrhlth1 Somhlth1 slfhlth1  
psqi neo day afraid1 unhappy1 annoyd1 ashmd1 guilty1 an-  
gry1 sad1 hostile1 scared1 upset1 irrtbl1 deprsd1 jttry1 drowsy1  
slugish1 worrid1 nervs1 lonely1 fatiged1 distrsd1 nPANAS1;  
USEVARIABLES = afraid1 scared1 nervs1 jttry1 guilty1  
ashmd1 irrtbl1 hostile1 upset1 distrsd1;  
CATEGORICAL = afraid1-distrsd1;  
**CLUSTER = subject day;**  
MISSING = all(999);  
TINTERVAL = day(1);  
ANALYSIS: **TYPE = CROSSCLASSIFIED RANDOM;**  
ESTIMATOR = BAYES;  
PROCESSORS = 2;  
BITERATIONS = (5000);  
THIN = 10;

---

---

MODEL:            % **WITHIN** %  
                  **na\_w BY afraid1-distrsd1\* (&1 1-10);**  
                  na\_w ON na\_w&1;  
                  % **BETWEEN SUBJECT** %  
                  **na\_subj BY afraid1-distrsd1\* (1-10);**  
                  na\_subj@1;  
                  % **BETWEEN DAY** %  
                  **na\_time BY afraid1-distrsd1\* (1-10);**

OUTPUT:           TECH1 TECH8 STDY STDYX TECH4 RESIDUAL  
                  FSCOMPARISON;

PLOT:             TYPE = PLOT3;  
                  FACTORS = ALL;


---

Run time: 54 minutes (dichotomized: 34 minutes)




# Results of Cross-Classified Factor Analysis with One Factor for 10 Ordinal NA Items

- $SD(\text{na}_w) = 0.81$
- $SD(\text{na}_{\text{subject}}) = 1.00$
- $SD(\text{na}_{\text{time}}) = 0.11$

The factor score plot for the `na_time` factor (on the between day level) shows a drop of 40% of the total factor SD over the 56 days:



# Posterior Distributions for the Factor Scores on Within, Between Subject, and Between Time



- How to use plots
- Why centering using latent variable decomposition into Within and Between avoids biases
- Why DIC has issues of non-comparability across certain models
- How standardization with random slopes and variances is done
- How to handle subjects
  - with no change over time (no within-subject variation);  $AR(1)=0$
  - with  $AR(1) \approx 1$
  - with spikes (a few high values surrounded by mostly low, flat values) - two-part modeling, mixture modeling?

# Upcoming Workshops and Talks on DSEM

- April 14: 1-day workshop at Utrecht University by Ellen Hamaker
- April 20: American Statistical Association Mental Health section webinar by us 3
- June 5-9: 5-day workshop at ICPSR (Univ of Michigan) by Bolger & Laurenceau
- June 29-30: Workshop on power at Penn State by Bolger & Laurenceau
- July 13: 1-day short course at Utrecht University by us 3
- July 14: 1-day Mplus Users' Meeting at Utrecht University
- July 17: 1-day pre-conference workshop at the International Meeting of the Psychometric Society in Zurich by us 3
- August 17-18: 2-day short course at Johns Hopkins University (preceded by a 1-day course on the RMA book) by us 3

- two-level modeling with random variances (UG ex 9.28, 9.29)
- two-level random autocorrelation modeling for short longitudinal data
- standardization for two-level models with random slopes and random variances
- random slopes for covariates with missing data
- new within/between scatter plots and histograms for two-level models, including sample and model-estimated cluster-specific means and variances
- new Posterior Predictive P-values for BSEM (Hojtink & van de Schoot, 2017)
- output in HTML format



# References

- Asparouhov, T. & Muthén, B. (2015). General random effect latent variable modeling: Random subjects, items, contexts, and parameters. In Harring, J. R., Stapleton, L. M., & Beretvas, S. N. (Eds.), *Advances in multilevel modeling for educational research: Addressing practical issues found in real-world applications*. Charlotte, NC: Information Age Publishing, Inc.
- Asparouhov, T., Hamaker, E.L. & Muthén, B. (2017). Dynamic structural equation models. Technical Report.
- Asparouhov, T., Hamaker, E.L. & Muthén, B. (2017). Dynamic latent class analysis. *Structural Equation Modeling*, 24, 257-269.
- Bolger, N. & Laurenceau, J-P. (2013). *Intensive longitudinal methods: An introduction to diary and experience sampling research*. New York: Guilford.
- Engle, R. & Watson, M. (1981). A one-factor multivariate time series model of metropolitan wage rates. *JASA*, 76, 774-781.
- Fox, J.P. (2010). *Bayesian item response theory*. Springer.
- Gelman, A., Carlin, J.B., Stern, H.S., & Rubin, D.B. (2014). *Bayesian data analysis*. Third edition. New York: Chapman & Hall.
- Granger, C.W.J. & Morris, M.J. (1976). Time series modelling and interpretation. *Journal of the Royal Statistical Society, Series A*, 139, 246-257.
- Hamaker, E.L. (2005). Conditions for the equivalence of the autoregressive latent trajectory model and a latent growth curve model with autoregressive disturbances. *Sociological Methods & Research*.
- Hamaker, E.L., Schuurman, N.K., & Zijlman, E.A.O. (2017). Using a few snapshots to distinguish mountains from waves: Weak factorial invariance in the context of trait-state research. *Multivariate Behavioral Research*.
- Hamaker, E.L. & Wichers, M. (2017). No time like the present: Discovering the hidden dynamics in intensive longitudinal data. *Current Directions in Psychological Science*.
- Hu, Nesselrode et al. (2016). Test reliability at the individual level. *Structural Equation Modeling*.

# References Continued

- Jak, S., Oort, F.J., & Dolan, C.V. (2013). A test for cluster bias: Detecting violations of measurement invariance across clusters in multilevel data. *Structural Equation Modeling*, 20, 265-282.
- Jak, S., Oort, F.J., & Dolan, C.V. (2014). Measurement Bias in Multilevel Data. *Structural Equation Modeling: A Multidisciplinary Journal*, 21:1, 31-39, DOI: 10.1080/10705511.2014.856694
- Lynch, S. M. (2010). *Introduction to applied Bayesian statistics and estimation for social scientists*. Springer.
- Molenaar, P. (1985). A dynamic factor model for the analysis of multivariate time series. *Psychometrika*, 50, 181-202.
- Muthén, B. (1994). Multilevel covariance structure analysis. In J. Hox & I. Kreft (eds.), *Multilevel Modeling, a special issue of Sociological Methods & Research*, 22, 376-398.
- Muthén, B. & Asparouhov, T. (2017). Recent methods for the study of measurement invariance with many groups. Alignment and random effects. Forthcoming in a special measurement invariance issue of *Sociological Methods & Research*.
- Muthén, B., Muthén, L., & Asparouhov, T. (2016). Regression and mediation analysis using Mplus. [www.statmodel.com](http://www.statmodel.com).
- Schuurman, N.K., Houtveen, J.H., & Hamaker, E.L. (2015). Incorporating measurement error in n=1 psychological autoregressive modeling. *Frontiers in Psychology*, 6, 1-15.
- Schuurman, N.K., Ferrer, E., de Boer-Sonnenschein, M., & Hamaker, E.L. (2016). How to compute cross-lagged associations in a multilevel autoregressive model. *Psychological Methods*, 21, 206-221.
- Shiyko et al. (2012). Using the time-varying effect model (TVEM) to examine dynamic associations between negative affect and self confidence on smoking urges. *Prevention Science*, 13, 288-299
- Shumway, R.H. & Stoffer, D.S. (2011). *Time series analysis and its applications*. New York: Springer.
- Wall, M.M., Guo, J., & Amemiya, Y. (2012). Mixture factor analysis for approximating a nonnormally distributed continuous latent factor with continuous and dichotomous observed variables. *Multivariate Behavioral Research*, 47, 276-313.
- Walls, T.D. & Schafer, J.L. (2006). *Models for intensive longitudinal data*. Oxford Univ Press.
- Wang, Hamaker, & Bergeman (2012). Investigating inter-individual differences in short-term intra-individual variability. *Psychological Methods*.